

I - Avant de commencer

a. Entrée des instructions

Toutes les instructions sont tapées devant un prompt rouge : `>`. Elles s'affichent en rouge. On termine son instruction obligatoirement par un point-virgule (`;`) si l'on veut que la réponse de MAPLE soit affichée, ou par deux points (`:`) si l'on veut que MAPLE exécute notre instruction sans l'afficher. Toutes les instructions entre deux prompts représentent une cellule.

♥ Changer de ligne sans valider

On presse en même temps sur `Shift ↑` et `Enter`.

Pour faire exécuter l'ensemble d'une expression, on positionne le curseur sur n'importe quelle ligne de la cellule et on presse la touche `Enter`. MAPLE répond en bleu.

♥ Vous avez perdu le curseur

Réduisez votre fenêtre puis maximisez-la à nouveau.

b. Aide en ligne

Pour obtenir de l'aide, il y a trois possibilités :

- Tapez la commande qui vous intéresse, par exemple `plot`, puis sélectionnez-la et tapez `F1`;
- Tapez `?plot`; puis `Enter`;
- Cliquez sur `Topic search` du menu `Help`.

c. Affectation

Pour stocker une valeur (un nombre, une fonction, un graphique, etc.) en mémoire, on l'affecte à un nom de variable (une étiquette) en utilisant le symbole `:=`.

Par exemple, si l'on veut affecter la valeur 2 à la variable « a », on tape :

```
a:=2;
```

et on obtient

Réponse du logiciel

```
a:=2
```

Pour ne pas afficher le résultat, on utilise `:` au lieu de `;` :

On peut évaluer toute fonction de la variable « a » :

```
a; a^2; cos(a); a/2;
```

On peut être amené à libérer une variable de son contenu. On utilise `unassign('variable')` :

```
unassign('a');  
a; a^2;
```

Pour libérer toutes les variables, on utilise `restart`.

d. Fonction et expression

On rentre une fonction comme on l'écrit. On utilise `-` et `>` pour faire `->`

```
f:=x->2*x+5;  
f(5);  
(f(a+h)-f(a))/h;
```

Il ne faudra pas confondre avec l'expression associée.

```
restart;
f:=2*x+1;
f(5);
(f(a+h)-f(a))/h;
```

Pour transformer une expression en fonction, on utilise :

`unapply(expression,variable)`

```
f:=2*x+1;
f(5);
f:=unapply(f,x);
f(5);
```

e. Opérations mathématiques de base

La multiplication s'obtient avec `*`, la division avec `/`, les puissances avec `^`, la racine carrée (Square Root en anglais...) avec `sqrt(nombre)`.

On peut de plus résoudre des équations avec `solve(équation,inconnue)`, calculer des limites avec `limit(expression,x=a,direction)` avec `a` un réel ou l'infini, `direction` prenant les valeurs `left` ou `right`.

On calcule des dérivées avec `diff(f(x),x)`, des intégrales avec `int(g(x),x)` ou `int(g(x),x=-a..b)`.

f. Résolutions d'équations

La commande magique est `solve(equation,variable)` :

```
solve(x^2-2*x+2,x);
solve(x^2+x+1,x);
solve(x^2+b*x+c,x);
solve(x^2+b*x+c,b);
```

Il peut cependant y avoir quelques problèmes :

```
solve(x^5+x^3-5*x+1,x);
solve(ln(x)=cos(exp(x)),x);
```

On peut cependant demander une valeur approchée des solutions avec `fsolve`.

Pour les systèmes, on utilise les mêmes commandes mais avec une syntaxe adaptée :

```
solve({x+y=1,x-y=2},{x,y});
```

II - Les graphes

a. En dimension 2

La commande à utiliser est `plot(expression,xmin..xmax,options)`. Les options sont nombreuses...

```
plot(sin(x),x=-4*Pi..5*Pi);
plot(sin(x),x=-4*Pi..5*Pi,y=-2..5,color=wheat,labels=['
abscisse','ordonnée'],title='le beau dessin');
```



L'accent grave

L'apostrophe utilisée s'obtient avec `[Alt Gr] + [7] + [Espace]`

Modifier la fenêtre peut être très utile. Par exemple, comment remédier à ce problème :

```
plot(tan(x),x=-2*Pi..2*Pi);
```

On peut superposer des graphes :

```
plot([sin(x),cos(x)],x=-3*Pi..2*Pi);
plot([sin(x),cos(x)],x=-3*Pi..2*Pi,color=[navy,wheat],
linestyle=[2,3]);
```

pour plus de précisions, allez voir l'aide de `plot`.

Pour des fonctions définies par morceaux, on utilise `piecewise`. Par exemple :

```
restart;
f:=x->piecewise(x<2,x*sin(x),x>=2,3*x^2+1);
plot(f(x),x=-4..4,y=-10..30,discont=true);
```

Certaines courbes sont définies implicitement par une équation qui ne correspond pas à une fonction. Par exemple, pour un cercle d'équation $x^2 + y^2 + x - 3 = 0$:

```
with(plots):
implicitplot(x^2+y^2+x-3=0,x=-5..5,y=-2..2,color=blue);
```

b. En dimension 3

Pour des fonctions de \mathbb{R}^2 dans \mathbb{R} , on utilise

```
plot3d(expression,x=xmin..xmax,y=ymin..ymax) :
```

```
plot3d(cos(x)*sin(y),x=-5..5,y=-5..5);
```

Un clic droit vous permet d'accéder à différentes options.

c. Animations

On peut créer une animation en créant une séquence de graphes dépendant d'un paramètre.

On va utiliser `seq(expression dépendant de a, valeurs prises par a)`

Par exemple :

```
g:=seq(plot(sin(a*x),x=-5..5),a=[-1,0,1,2,3,4,5]):
```

Pour afficher l'animation, on doit aller chercher la commande `display` dans la bibliothèque `plots` qui n'est pas chargée par défaut avec la commande `with` et lui indique l'option `insequence=true` pour ne pas afficher tous les graphes en même temps :

```
with(plots):
display(g,insequence=true);
```

Vous cliquez sur le graphe et une barre d'icône apparaît pour lancer l'animation.

Il n'y a pas assez de valeurs de a. On peut en créer toute une série avec `seq` :

```
g:=seq(plot(sin(a*x),x=-5..5),a=[seq(-1+0.01*h,h=0..100)])
:
```

III - Quelques erreurs types

Voici une série d'erreurs que l'on retrouve souvent : il faut savoir les corriger seul(e)...

Partons d'un code correct :

```
f:=x->sin(x)/x;
a:=20;
plot(f(x),x=-a..a);
```

Ça marche.

Mais

```
restart;
f:=x->sin(x)/x;
a=20;
plot(f(x),x=-a..a);
```

ne marche plus : pourquoi ?

Et pour ceci :

```
restart;
f:=x->sin(x)/x;
a:=20;
plot(f(x),x=-a..a);
```

Le message d'erreur est différent mais c'est le même type d'erreur.

La plus fréquente des erreurs :

```
a:=20
```

à égalité avec celle-ci :

```
restart;
f:=x->sin(x)/x;
a:=20;
plt(f(x),x=-a..a);
```

Sans oublier celle-ci :

```
restart;
f:=x->sin(x)/x;
a:=20;
plot(,x=-a..a);
```

IV - Exercices

Exercice 1

Regardez l'aide pour `evalf` et calculez π avec une précision de 1000 chiffres.

Exercice 2

Explorez les fonctions $f : x \mapsto x^4 - 2 * x^2 + 3$, $g : x \mapsto \sqrt{\frac{x^2 - 2x + 2}{x^2 + 4}}$, $h : x \mapsto \arcsin(x) - \ln(x)$: limites, dérivées, signe de la dérivée, graphe...

♥ Tableau de valeurs

On peut procéder ainsi :

```
f:=x->1/(x-1/2);
valx:=[1,2,3,5,10,15,20,100]:
valfx:=seq([x,f(x)],x=valx):
titre:=['x','f(x)']:
array([titre,valfx]);
```

sachant que `array(liste)` crée un tableau.

Définissez également un tableau de valeurs pour les trois fonctions.

Exercice 3

La position d'un mobile en mouvement harmonique amorti est donné par l'équation :

$$x(t) = e^{-0,1t} \left(\frac{2}{3} \sin(10t) + \frac{4}{5} \cos(10t) \right)$$

Déterminez les trois premiers instants où la vitesse du corps est nulle.

Exercice 4

Trouvez les constantes a et b qui rendent la fonction suivante continue sur \mathbb{R} à l'aide de MAPLE :

$$f(x) = \begin{cases} x^2 + bx + 1 & \text{si } x < 5 \\ 8 & \text{si } x = 5 \\ ax + 3 & \text{si } x > 5 \end{cases}$$

Vous utiliserez `piecewise`, `solve`, `limit`. Pour avoir la limite à gauche en x_0 on entre `limit(f(x),x=x0,left)`.

Exercice 5

Trouvez les constantes a , b et c telles que le graphe de $f : x \mapsto 3x^4 + ax^3 + bx^2 + cx + d$ ait des tangentes horizontales en $(2; -3)$ et $(0; 7)$. Tracez le graphe associé. Il y a un troisième point où la tangente est horizontale : trouvez-le ! Vérifiez ensuite s'il s'agit de maxima ou de minima relatifs ou ni l'un ni l'autre.

Exercice 6

Parmi les exercices d'analyse faits en classe, reprenez ceux où MAPLE aurait pu vous aider.

Exercice 7

On considère la fonction $f : x \mapsto x^3 - 2x^2 + 1$. Créez une animation permettant de comparer différentes sécantes passant par le point de coordonnées $(2, f(2))$ et la tangente à \mathcal{C}_f en ce même point.